
The Book of V

The goal of this book is to teach the V language from the beginning. This book
is a work in progress and I’ll be editing and including new chapters as I too
discover more about this new language. The book versions are released as
separate Git [https://www.git-scm.com/] tags containing assets hopefully
satifying everyone’s taste in reading formats.

See something incorrectly described, buggy or want to contribute a chapter of
your own? Feel free to send a pull request and we will find a way to include
it!

Feel free to buy me a coffee [image: donate] [https://paypal.me/peterbadida]

Installation


Prerequisities


	Internet connection


	Git (2.17.1)


	Docker (19.03.2)





Note

These are versions are mine, but older ones should work too for simple
cloning and building Docker image.






Cloning, compiling and building

To ensure the same environment everywhere we’ll use Docker and the default
Dockerfile from the V repository [https://github.com/vlang/v]. Current version is fetched this commit [https://github.com/vlang/v/tree/2880baa1bc2730519f3ce01e5f18b7a4363206b8/].

#same container that golang use
FROM buildpack-deps:buster-curl

LABEL maintainer="ANAGO Ronnel <anagoandy@gmail.com>"
WORKDIR /opt/vlang
RUN apt-get update && \
    DEBIAN_FRONTEND=noninteractive apt-get install -y --no-install-recommends gcc clang make git && \
    apt-get clean && rm -rf /var/cache/apt/archives/* && \
    rm -rf /var/lib/apt/lists/*
COPY . .
RUN make && \
    ln -s /opt/vlang/v /usr/local/bin/v

CMD [ "bash" ]





Download it with cloning the repo first (currently b51b885):

git clone https://github.com/vlang/v
# or
git clone git@github.com:vlang/v





Then proceed to building the Docker image locally. During the building process
one of the instructions is make which compiles V in that Docker image. Once
it’s built, create a container and enter it:

docker build --tag vlang .
docker run --interactive --tty vlang





Your console should now look similar to this:

root@16b5a9d05074:/opt/vlang#





The environment you entered is an isolated part of your system which contains
a V installation:

root@16b5a9d05074:/opt/vlang# v --version
V 0.1.21 b51b885





By default the whole environment is isolated, but that prevents us from adding
or editing files from an editor that’s not installed in the Docker image. For
that we will use mounting of a host directory into the container so we can:


	Edit the files on the OS with an editor of own choice


	Compile and run them in a consistent and reproducible environment




We will also use shortened flags for docker instead of writing e.g.
--interactive in full.

docker run -i -t -v $(pwd):/opt/src -w /opt/src vlang





This command will run the Docker container from vlang image in an
interactive mode, will allocate a tty for it, make visible
the directory you are in (pwd) to the container at location /opt/src
and change the default working directory to the project location: /opt/src.


Note

Once you have built the Docker image, you can navigate to any folder on your
computer and run the command above. This is helpful if you have multiple
projects because it’ll bring the consistent environemnt with you wherever
you go.

In case you make some changes to the Dockerfile, it’s nice to have it
always available even between multiple machines. You can do that with
pushing and pulling Docker images [https://docs.docker.com/ee/dtr/user/manage-images/pull-and-push-images/].

Modes

V provides multiple modes of running while the default (v or v runrepl)
throws you into the REPL (Read-Execute-Print-Loop) console where you can try
the language or write small scripts.


REPL

Although not in the way as it is for Python or JavaScript, this
interpreter-like mode takes your code and in the background writes it to
a temporary .v file which is then compiled and run. V then returns the
output back to you in the console.

v  # or v runrepl





root@16b5a9d05074:/opt/vlang# v
V 0.1.21 b51b885
Use Ctrl-C or `exit` to exit
>>>





The console provides a simple help command that lists all available console
commands:

>>> help

V 0.1.21 b51b885
  help                   Displays this information.
  Ctrl-C, Ctrl-D, exit   Exits the REPL.
  clear                  Clears the screen.






Note

Currently there is a hidden step between compiling .v file and running
the final program. V requires a C compiler present on the system and
attempts to compile .v file to .c which is then compiled to machine
code a CPU understands. After that V runs the binary produced by the C
compiler and retrieves output back.




Warning

Currently there is no support for command history, therefore arrows,
Control-P, Alt-P or any combination of them will result in an escape code
printed to the console.

>>> ^[[A
>>> ^P
>>> ^[p
>>> ^[^P
>>> ^[[1;5A
>>> ^[[1;3A







By default you have an access to the builtin [https://github.com/vlang/v/tree/master/vlib/builtin] module

>>> println("Hello, world!")
Hello, world!

// or even

>>> print("Hello, world!")
Hello, world!





In its simplest form it can be used as a calculator:

>>> print(1 + 1)
2





For more examples check Calculator test cases.




Compiler

V executable when provided with an argument that contains a .v suffix will
open that file, compile it and produce a same-named binary.

// helloworld.v
println("Hello, world!")  // with a new like character \n
print("Hello, world!")    // without a new like character





Compile with:

v helloworld.v
./helloworld






Note

In case you can’t execute the output file try changing the file into
an executable with chmod +x <file>.






Runner

Similar to REPL mode, this mode in the background compiles and attempts to run
your file.

v run helloworld.v

Chapter I: Calculator

For the first program (except Hello world of course!) let’s create a simple CLI
calculator. We start with a clean file in your favorite text editor:

// calc.v






Operations

We’ll start with the addition. In V a function is created by fn
keyword and a basic numeric type “integer”
is noted as int:

// calc.v
fn add(left int, right int) int {
    return left + right
}





If we run the file:

v run calc.v





nothing happens because we are missing the entrypoint for the executable,
or simply said, a place where computer should begin the execution of a program.
In V the entrypoint is known as a main function declared as fn main().
Let’s add a main() function that adds two integers and displays the output
in the console. For that we’ll use a built-in println(s string) [https://vlang.io/docs#println] function:

// calc.v
fn add(left int, right int) int {
    return left + right
}

fn main() {
    println(add(1, 2))
}





Now we can see the output. Let’s include the rest of +-*/ operations
in the same way we created add() function:

// calc.v
fn add(left int, right int) int {
    return left + right
}

fn sub(left int, right int) int {
    return left - right
}

fn mul(left int, right int) int {
    return left * right
}

fn div(left int, right int) int {
    return left / right
}

fn main() {
    println(add(1, 2))
    println(sub(1, 2))
    println(mul(1, 2))
    println(div(1, 2))
}





As we call each of the functions to perform a basic operation, we will see
this output. Noticed something strange?

3
-1
2
0





1 / 2 is apparently 0. But why? Shouldn’t the output be 0.5?

Yes and no. For now we use int type which does not allow
a decimal mark and values after that symbol. There is a different
type that allows it written as f32 (short for 32-bit
floating-point number).

Switch all types from int to f32 and run the program again.

// calc.v
fn add(left f32, right f32) f32 {
    return left + right
}

fn sub(left f32, right f32) f32 {
    return left - right
}

fn mul(left f32, right f32) f32 {
    return left * right
}

fn div(left f32, right f32) f32 {
    return left / right
}

fn main() {
    println(add(1, 2))
    println(sub(1, 2))
    println(mul(1, 2))
    println(div(1, 2))
}





3.000000
-1.000000
2.000000
0.500000





Nice, we can see a proper result for 1 / 2 operation.




Input

We can see the program computing results, but it’s only for the hard-coded
values directly in the calc.v file. This way we’d always need to rewrite
our calculator.

As you’ve already noticed when running a V program, V already knows what file
you want to use by you providing a filename in the console. The same way V
can use your value for compiling we can use it for computing results. The input
values can be fetched with the help of os [https://github.com/vlang/v/tree/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/os] module.

In V you can use a module by importing it via import
keyword. From that module we will need a constant
os.args [https://github.com/vlang/v/blob/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/os/os.v#L30] that returns an array of another kind of V
type - string [https://vlang.io/docs#strings].

import os

fn add(left f32, right f32) f32 {
    return left + right
}

fn sub(left f32, right f32) f32 {
    return left - right
}

fn mul(left f32, right f32) f32 {
    return left * right
}

fn div(left f32, right f32) f32 {
    return left / right
}

fn main() {
    println(add(1, 2))
    println(sub(1, 2))
    println(mul(1, 2))
    println(div(1, 2))
    println(os.args)
}





Now you should see even the array of arguments passed to the calculator
program the same as below:

3.000000
-1.000000
2.000000
0.500000
["./calculator-basic-ops-float"]





The first argument will always be a name of the executable that’s running,
which in this case means ./calc, and anything other added after the path
to the executable is added as the next argument.

# call
v run calc.v argument
./calc argument

# output
...
["./calc", "argument"]








Pseudo-stack with array [https://vlang.io/docs#arrays]

Once we can access the console arguments, we can quickly implement so called
Reverse Polish notation [https://en.wikipedia.org/wiki/Reverse_Polish_notation] with a for loop [https://vlang.io/docs#for], if conditional [https://vlang.io/docs#if] and stack [https://en.wikipedia.org/wiki/Stack_%28abstract_data_type%29].

First we skip the executable path:

import os

fn main() {
    for idx, value in os.args {
        if idx == 0 {
            continue
        }
        println(value)
    }
}





To implement Reverse Polish notation [https://en.wikipedia.org/wiki/Reverse_Polish_notation] we will use an array [https://vlang.io/docs#arrays] as a pseudo-stack [https://en.wikipedia.org/wiki/Stack_%28abstract_data_type%29]
structure for storing the f32. To create a variable V uses a simple
<name> := <value> syntax e.g. number := 1, however for an array there is
a catch. We need to go a little bit further and specify the type of all values
in it as <name> := []<type>.

fn main() {
    stack := []f32
    stack << 1.2  // error
}





After we use a keyword mut, we mark the variable as
editable and can use << operator for the array to append a new value to it.

Currently there is no quick way for popping the last element from an array
while removing it at the same time, therefore we will access the last element
by its position in angle brackets (array[idx]) in the array and then use
array.delete() [https://github.com/vlang/v/blob/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/builtin/array.v#L107] to remove it.

fn main() {
    mut stack := []f32

    // push
    stack << 1.2
    stack << 2.2
    stack << 3.2

    // print the array and its length
    println(stack)
    println(stack.len)

    // pop last item
    temp := stack[stack.len - 1]
    stack.delete(stack.len - 1)

    // print the array and the popped item
    println(stack)
    println(temp)
}





As you can see, an array [https://vlang.io/docs#arrays] has an array.len attribute [https://github.com/vlang/v/blob/b73387647cdc1445a9c8a0ea000d8f1fae4d02f8/vlib/builtin/array.v#L14]. It’s changed on each
resizing manipulation of array [https://vlang.io/docs#arrays].

Back to the calculator code, we will use this pseudo-stack [https://en.wikipedia.org/wiki/Stack_%28abstract_data_type%29] with
manually pushing and removing items to implement Reverse Polish notation [https://en.wikipedia.org/wiki/Reverse_Polish_notation] from console arguments.

We’ll create two array “buckets” for two categories of operators according to
their precedence in an ordinary calculator.

After that let’s take care of an obvious error that might be raised - calling
an operator function when there isn’t enough values on the stack. We need to
check the number of elements in the stack array by its len attribute
as we did for popping the values from it in previous example and then exit the
program with a warning for which we’ll use panic(s string) [https://github.com/vlang/v/blob/b73387647cdc1445a9c8a0ea000d8f1fae4d02f8/vlib/builtin/builtin.v#L110].

We can concat a variable, to a different string by using $ symbol and
a name of a variable in a string like this: println("Value: $my_variable").

fn main() {
    // RPN stack
    stack := []f32
    prioritized := ["mul", "div"]
    normal := ["add", "sub"]

    for idx, value in os.args {
        if idx == 0 {
            continue
        }

        println(value)
        if (value in prioritized || value in normal) && stack.len < 2 {
            panic("No values to use for operator $value")
        }
    }
}





Now let’s append the console arguments on the stack if they are not one of
the operator functions’ names we added to the arrays. By default any value from
the console arguments is a string [https://vlang.io/docs#strings] which means we are missing one step.
We need to convert the value to f32 before appending it. By looking at the
string implementation [https://github.com/vlang/v/blob/333f0ab89f22846e1023d1ab6b1deec5fde592ad/vlib/builtin/string.v#L46] we can find its function for conversion string.f32(s string) [https://github.com/vlang/v/blob/333f0ab89f22846e1023d1ab6b1deec5fde592ad/vlib/builtin/string.v#L193] which we
should use before trying to append the value on the stack.

Once the values are converted and on the stack, we need to check if there are
at least two and in the next console argument is an operator function’s name
pop the values into left and right variables which will be used for
printing out the result.

fn main() {
    // RPN stack
    mut stack := []f32
    prioritized := ["mul", "div"]
    normal := ["add", "sub"]

    for idx, value in os.args {
        if idx == 0 {
            continue
        }

        if (value in prioritized || value in normal) {
            if stack.len < 2 {
                panic("No values to use for operator $value")
            } else {
                right := stack[stack.len - 1]
                stack.delete(stack.len - 1)

                left := stack[stack.len - 1]
                stack.delete(stack.len - 1)

                if value == "add" {
                    println(add(left, right))
                } else if value == "sub" {
                    println(sub(left, right))
                } else if value == "mul" {
                    println(mul(left, right))
                } else if value == "div" {
                    println(div(left, right))
                } else {
                    println("$left $value $right")
                    println(stack)
                    panic("This should not happen!")
                }
                continue
            }
        }

        if (value in prioritized || value in normal) && stack.len < 2 {
            panic("No values to use for operator $value")
        }

        if !(value in prioritized || value in normal) {
            stack << value.f32()
        }
    }
}





Now we can check some basic instructions for our calculator this way:

// v calc.v
// ./calc 1.5 2 add
3.500000
// ./calc 1.5 2 sub
-0.500000
// ./calc 1.5 2 mul
3.000000
// ./calc 1.5 2 div
0.750000





Although usable, it’s very limited as it does not provide any option for joined
operations such as (1 + 2) * (3 / (4 + 5)). First we need to convert this
operation into Reverse Polish notation [https://en.wikipedia.org/wiki/Reverse_Polish_notation] instructions so we can appropriately edit the main()
function.

(1 + 2) * (3 / (4 + 5)) = 1
1 2 + 3 4 5 + / *
1 2 add 3 4 5 add div mul

// each value is added on the stack as present
// computing starts immediately after an operator is present and the two
// closest values on the stack are popped in the reversed order i.e.
// right first, left second
1
1 2
1 2 +
3
3 3
3 3 4
3 3 4 5
3 3 4 5 +
3 3 9
3 3 9 /
3 1/3
3 1/3 *
1
= 1





For that we need to rework the result handling a bit and put it on stack
instead of printing out right away - switch println(operator(left, right))
to stack << operator(left, right) and then, after the computation is done,
make sure there are no console arguments remaining. Then print the whole stack
back to the console.


Note

Optimal result is having only a single element present on the stack, however
it can happen that there will be an additional result if we provide more
values than operators + 1.



Here is the complete solution. Obviously it can be optimized and refactored
further, but that I’ve kept for you to have fun.

import os

fn add(left f32, right f32) f32 {
    return left + right
}

fn sub(left f32, right f32) f32 {
    return left - right
}

fn mul(left f32, right f32) f32 {
    return left * right
}

fn div(left f32, right f32) f32 {
    return left / right
}

fn main() {
    // RPN stack
    mut stack := []f32
    prioritized := ["mul", "div"]
    normal := ["add", "sub"]

    for idx, value in os.args {
        if idx == 0 {
            continue
        }

        if (value in prioritized || value in normal) {
            if stack.len < 2 {
                panic("No values to use for operator $value")
            } else {
                right := stack[stack.len - 1]
                stack.delete(stack.len - 1)

                left := stack[stack.len - 1]
                stack.delete(stack.len - 1)

                if value == "add" {
                    stack << add(left, right)
                } else if value == "sub" {
                    stack << sub(left, right)
                } else if value == "mul" {
                    stack << mul(left, right)
                } else if value == "div" {
                    stack << div(left, right)
                } else {
                    println("$left $value $right")
                    println(stack)
                    panic("This should not happen!")
                }

                if idx == os.args.len - 1 {
                    println(stack)
                    exit(0)
                } else {
                    continue
                }
            }
        }

        if (value in prioritized || value in normal) && stack.len < 2 {
            panic("No values to use for operator $value")
        }

        if !(value in prioritized || value in normal) {
            stack << value.f32()
        }
    }
}

Chapter II: Hangman

After the first chapter you should know how to use keywords
fn and mut, types int, f32, string [https://vlang.io/docs#strings],
array [https://vlang.io/docs#arrays] as well as converting string [https://vlang.io/docs#strings] input to f32 and some
basic array [https://vlang.io/docs#arrays] operations with <<, array.len attribute [https://github.com/vlang/v/blob/b73387647cdc1445a9c8a0ea000d8f1fae4d02f8/vlib/builtin/array.v#L14] and array.delete() [https://github.com/vlang/v/blob/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/builtin/array.v#L107].


Game loop

In this chapter we will create a Hangman game that will pick a random word from
a predefined array of values. As for any common game, we need to create a loop
in which we check the inner state of the game and request a user input or
action if necessary.

fn game_loop() {}

fn main() {
    game_loop()
}





To define what should be present in the game loop we need to check the
Hangman description [https://en.wikipedia.org/wiki/Hangman_(game)#Overview]. So according to that, the game loop should have a player
always guessing either a letter or the whole hidden word until a player makes
6 mistakes. In between the guesses the game should visibly note player’s
mistakes and unfold all places of a guessed letter. The game ends either by
providing a correct word, guessing all letters correctly or at player’s 6th
mistake.

Let’s define the game ending conditions first as having 6 user attempts total,
wrap checking for the user attempt in an infinite loop and then stopping the
game loop once the conditions are not matching. An infinite loop in V is
defined as a for loop [https://vlang.io/docs#for] without any specified condition. Afterwards the loop
can be broken by a break keyword.

fn game_loop() {
    max_attempts := 6
    mut attempts := 0
    for {
        attempts++
        println("User attempt $attempts")
        if attempts >= max_attempts {
            break
        }
    }
}

fn main() {
    game_loop()
    println("Game over")
}





We can see that the amount of attempts is a value that won’t change, therefore
we can declare it as a constant with const keyword.
Unlike ordinary variables a value to a const is assigned via = instead
of := and can’t be changed.

const (
    max_attempts = 6
)

fn game_loop() {
    mut attempts := 0
    for {
        attempts++
        println("User attempt $attempts")
        if attempts >= max_attempts {
            break
        }
    }
}

fn main() {
    game_loop()
    println("Game over")
}





Next we retrieve a user input so we can. If the length of the input is 1 we
will treat it as guessing a character from the whole word and display all its
occurences if the character is present. If the input is longer than one
character, player is guessing the whole word and only if the word matches we
display it. For any other case just take it as a failed attempt.

For user input import os [https://github.com/vlang/v/tree/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/os], then use its os.get_line() [https://github.com/vlang/v/blob/fdfa564865331a2bda077164d804d6dc90b83498/vlib/os/os.v#L540] function to
retrieve a single line from console - or in other words an input terminated by
a single Enter key [https://en.wikipedia.org/wiki/Enter_key].

import os

const (
    max_attempts = 6
)

fn game_loop() {
    mut attempts := 0
    guess_word := "hangman"
    mut user_input := ""

    for {
        attempts++
        println("User attempt $attempts")

        user_input = os.get_line()
        if user_input.len > 1 {
            println("Guessing a word: $user_input")
        } else if user_input.len == 1 {
            println("Guessing a character: $user_input")
        }

        if attempts >= max_attempts {
            break
        }
    }
}

fn main() {
    game_loop()
    println("Game over")
}





Once we have the input available, let’s add a sample word hangman to
a variable. Then create a mask of that word, a value constructed of -
characters in the same length as the guess word. That’s easily achievable with
string.repeat(count int) [https://github.com/vlang/v/blob/7696a600f6fc978ec655feea6d54a14c0d38caab/vlib/builtin/string.v#L1077] function.

import os

const (
    max_attempts = 6
)

fn game_loop() {
    mut attempts := 0
    guess_word := "hangman"
    display_word := "-".repeat(guess_word.len)
    mut user_input := ""

    for {
        attempts++
        println("Attempt $attempts")
        println("Word: [ $display_word ]")

        user_input = os.get_line()
        if user_input.len > 1 {
            println("Guessing a word: $user_input")
        } else if user_input.len == 1 {
            println("Guessing a character: $user_input")
        }

        if attempts >= max_attempts {
            break
        }
    }
}

fn main() {
    game_loop()
    println("Game over")
}





If a player guesses correctly, go through the variable which stores hangman
word, find each occurence of the character and replace the - characters
with uncovered ones according to the position in the original word.

import os

const (
    max_attempts = 6
)

fn game_loop() {
    mut attempts := 0
    guess_word := "hangman"
    mut display_word := "-".repeat(guess_word.len)
    mut user_input := ""

    for {
        attempts++
        println("Attempt $attempts")
        println("Word: [ $display_word ]")

        user_input = os.get_line()
        if user_input.len > 1 {
            println("Guessing a word: $user_input")
            if user_input == guess_word {
                display_word = guess_word
            }
        } else if user_input.len == 1 {
            println("Guessing a character: $user_input")
            mut buffer := ""
            for idx, value in guess_word {
                if value == display_word[idx] {
                    buffer += display_word[idx].str()
                    continue
                }

                if value != user_input[0] {
                    buffer += "-"
                    continue
                }

                buffer += guess_word[idx].str()
            }
            display_word = buffer
        }

        if display_word == guess_word {
            println("Correctly guessed!")
            break
        }

        if attempts >= max_attempts {
            println("Game over")
            break
        }
    }
}

fn main() {
    game_loop()
}





Notice the sections where the str() function is called. While a word is
stored as a string [https://vlang.io/docs#strings], that’s in simple terms just an array [https://vlang.io/docs#arrays] of
byte types. A byte on the other hand is so similar to an
int that the str() function is the same for both - int.str(c byte) [https://github.com/vlang/v/blob/14c273f273c2bd8bff1d0cb2f775ec755e43455c/vlib/builtin/int.v#L180].




Improvements

If we look properly at this large game loop, we can see multiple parts that
can be pulled out into separate functions which will increase the readability
of the overall code. Let’s move the code working with user input and name it as
guess() function. This function will take multiple string [https://vlang.io/docs#strings]
parameters and also return one string [https://vlang.io/docs#strings]. We reflect those properties to
the function declaration and the result should look like this:
fn guess(input string, word string, mask string) string.

fn guess(input string, word string, mask string) string {
    mut new_mask := mask

    if input.len > 1 {
        if input == word {
            new_mask = word
        }
    } else if input.len == 1 {
        mut buffer := ""
        for idx, value in word {
            if value == mask[idx] {
                buffer += mask[idx].str()
                continue
            }

            if value != input[0] {
                buffer += "-"
                continue
            }

            buffer += word[idx].str()
        }
        new_mask = buffer
    }
    return new_mask
}





For the conditions of word matching and attempts we create check_continue()
function and make it return a bool type so the for loop [https://vlang.io/docs#for]
can automatically check the value and exit game loop when necessary. Similarly
for the match of the guessed word and already uncovered letters we create
check_win() function so the logic is kept at one place.

fn check_win(word string, mask string) bool {
    return word == mask
}


fn check_continue(word string, mask string, attempts int) bool {
    return !check_win(word, mask) && attempts < max_attempts
}





Finally, we move out the game summary prints to print_summary() function,
clean unnecessary variables and add spacing between code lines.

import os

const (
    max_attempts = 6
)

fn guess(input string, word string, mask string) string {
    mut new_mask := mask

    if input.len > 1 && input == word {
        new_mask = word
    } else if input.len == 1 {
        mut buffer := ""

        for idx, value in word {
            if value == mask[idx] {
                buffer += mask[idx].str()
                continue
            }

            if value != input[0] {
                buffer += "-"
                continue
            }

            buffer += word[idx].str()
        }
        new_mask = buffer
    }
    return new_mask
}


fn check_win(word string, mask string) bool {
    return word == mask
}


fn check_continue(word string, mask string, attempts int) bool {
    return !check_win(word, mask) && attempts < max_attempts
}


fn print_summary(word string, mask string) {
    if check_win(word, mask) {
        println("Correctly guessed!")
    } else {
        println("Game over!")
    }
}


fn game_loop() {
    guess_word := "hangman"
    mut display_word := "-".repeat(guess_word.len)

    mut attempts := 0
    mut do_loop := true

    for do_loop {
        attempts++
        println("Attempt $attempts")
        println("Word: [ $display_word ]")

        display_word = guess(os.get_line(), guess_word, display_word)
        do_loop = check_continue(guess_word, display_word, attempts)
    }

    print_summary(guess_word, display_word)
}

fn main() {
    game_loop()
}





While the current state of the game is workable, we still need to make it
dynamic otherwise the game would be always the same and after the first try
everyone knows the answer. We can solve this by using a file named
words.txt which will contain guess words one per each line.

To read lines from a file on your operating system use os.read_lines(path string) [https://github.com/vlang/v/blob/fdfa564865331a2bda077164d804d6dc90b83498/vlib/os/os.v#L142]
function. Specify just the name of the file to open it from the current
“working” directory or in other words, from the folder you run your program
from.

After the array [https://vlang.io/docs#arrays] of string [https://vlang.io/docs#strings] s is retrieved we need to pull
a single word for the game to begin. Import rand [https://github.com/vlang/v/tree/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/rand] and use rand.next(max int) [https://github.com/vlang/v/blob/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/rand/rand.v#L11]
from it. Make sure the maximum value is set to the length of the words
array [https://vlang.io/docs#arrays] to only access its item by an index within the range of
available words.

fn load_word(path string) string {
    lines := os.read_lines(path)
    return lines[rand.next(lines.len)]
}





If you try to include this function into the already existing game, it will
most likely have a consistent behavior (always the same word chosen). That’s
because a randomness seed needs to be different on each run. For that import
time [https://github.com/vlang/v/tree/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/time] and input time.now() [https://github.com/vlang/v/blob/1ac162635781af7069c6490d821d8b05aea87c5b/vlib/time/time.v#L77] to a call setting the seed - rand.seed(s int) [https://github.com/vlang/v/blob/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/rand/rand.v#L11].

Since rand.seed(s int) [https://github.com/vlang/v/blob/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/rand/rand.v#L11] requires an int type we can convert the
time.Time [https://github.com/vlang/v/blob/1ac162635781af7069c6490d821d8b05aea87c5b/vlib/time/time.v#L47] into a UNIX timestamp which is a number we can safely use for
seeding in this particular case. time.Time [https://github.com/vlang/v/blob/1ac162635781af7069c6490d821d8b05aea87c5b/vlib/time/time.v#L47] stores it in time.Time.uni attribute [https://github.com/vlang/v/blob/1ac162635781af7069c6490d821d8b05aea87c5b/vlib/time/time.v#L55].

Once the seed is set we can call rand.next(max int) [https://github.com/vlang/v/blob/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/rand/rand.v#L11]. Now change the hard-coded
guess word in game_loop() into load_word("words.txt") and create a file
named words.txt in the same folder as is the hangman .v file. You can
find a sample file in the Appendix section.

import os
import rand
import time


const (
    max_attempts = 6
)


fn load_word(path string) string {
    lines := os.read_lines(path)
    rand.seed(time.now().uni)
    return lines[rand.next(lines.len)]
}


fn guess(input string, word string, mask string) string {
    mut new_mask := mask

    if input.len > 1 && input == word {
        new_mask = word
    } else if input.len == 1 {
        mut buffer := ""

        for idx, value in word {
            if value == mask[idx] {
                buffer += mask[idx].str()
                continue
            }

            if value != input[0] {
                buffer += "-"
                continue
            }

            buffer += word[idx].str()
        }
        new_mask = buffer
    }
    return new_mask
}


fn check_win(word string, mask string) bool {
    return word == mask
}


fn check_continue(word string, mask string, attempts int) bool {
    return !check_win(word, mask) && attempts < max_attempts
}


fn print_summary(word string, mask string) {
    if check_win(word, mask) {
        println("Correctly guessed!")
    } else {
        println("Game over!")
    }
}


fn game_loop() {
    guess_word := load_word("words.txt")
    mut display_word := "-".repeat(guess_word.len)

    mut attempts := 0
    mut do_loop := true

    for do_loop {
        attempts++
        println("Attempt $attempts")
        println("Word: [ $display_word ]")

        display_word = guess(os.get_line(), guess_word, display_word)
        do_loop = check_continue(guess_word, display_word, attempts)
    }

    print_summary(guess_word, display_word)
}

fn main() {
    game_loop()
}

Chapter III: Word counter

Counting can be a very simple example which can help explaining the grouping
of some variables under one roof. Let’s jump straight into fetching console
arguments via os [https://github.com/vlang/v/tree/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/os] and define three modes this program will work with:


	-w or --words


	-l or --lines


	-c or --chars




import os


fn parse_mode(args []string) string {
    mut mode := ""
    words := ["-w", "--words"]
    lines := ["-l", "--lines"]
    chars := ["-c", "--chars"]

    if args[1] in words {
        mode = "words"
    } else if args[1] in lines {
        mode = "lines"
    } else if args[1] in chars {
        mode = "chars"
    }
    return mode
}


fn main() {
    mode := parse_mode(os.args)
    println("Mode: $mode")
}





To generalize the mode we create a container for it - a struct [https://vlang.io/docs#structs] - with
struct keyword. The container will hold the mode’s
name, its console arguments and later some other attributes. By default
everything stored in a struct [https://vlang.io/docs#structs] is immutable and pretty much inaccessible
which allows us to have efficient abstractions without hacky hot-fixes unless
we explicitly allow them.

struct Mode {
    name string
    cli_args []string
}





Once the struct is declared with name, curly brackets and its attributes it’s
ready for usage. Although there are multiple ways for populating its attributes
with values, we’ll use explicitly stated attribute names before values. This
allows us to specify the attributes in an unordered way and it’s quite
important in the long run from the project maintainability perspective due to
no requirement of keeping the order of the struct [https://vlang.io/docs#structs] attributes (which may
change by including a new feature).

import os

struct Mode {
    name string
    cli_args []string
}


fn parse_mode(args []string) Mode {
    mut mode := Mode{}
    words := Mode{name: "words", cli_args: ["-w", "--words"]}
    lines := Mode{name: "lines", cli_args: ["-l", "--lines"]}
    chars := Mode{name: "chars", cli_args: ["-c", "--chars"]}

    if args[1] in words.cli_args {
        mode = words
    } else if args[1] in lines.cli_args {
        mode = lines
    } else if args[1] in chars.cli_args {
        mode = chars
    }
    return mode
}


fn main() {
    mode := parse_mode(os.args)
    println("Mode: $mode.name")
}





Now we can propagate each struct [https://vlang.io/docs#structs] that’s initialized out of the function.
Although V presents itself as not having a global space for symbols,
const keyword [https://vlang.io/docs#consts] creates a very similar space as global one, but on the module level.
Since we can’t rewrite the value assigned as a constant, in combination with
import keyword the module space is quite a powerful and
safe feature.

import os

struct Mode {
    name string
    cli_args []string
}


const (
    words = Mode{name: "words", cli_args: ["-w", "--words"]}
    lines = Mode{cli_args: ["-l", "--lines"], name: "lines"}
    chars = Mode{name: "chars", cli_args: ["-c", "--chars"]}
)


fn parse_mode(args []string) Mode {
    mut mode := Mode{}

    if args[1] in words.cli_args {
        mode = words
    } else if args[1] in lines.cli_args {
        mode = lines
    } else if args[1] in chars.cli_args {
        mode = chars
    }
    return mode
}


fn main() {
    mode := parse_mode(os.args)
    println("Mode: $mode.name")
}





Each mode needs some kind of configuration so the counter knows what to use for
distinguishing between words, lines or characters. This configuration we name
sep as in separator and set it to <space> for words, \n for
lines and <empty string> for characters. Note that the last one will count
even <space> or \n as a character.

struct Mode {
    name string
    cli_args []string
    sep string
}


const (
    words = Mode{name: "words", cli_args: ["-w", "--words"], sep: " "}
    lines = Mode{cli_args: ["-l", "--lines"], name: "lines", sep: "\n"}
    chars = Mode{name: "chars", cli_args: ["-c", "--chars"], sep: ""}
)





To count we need to fetch the path from os.args [https://github.com/vlang/v/blob/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/os/os.v#L30], open the file and process
its contents with a counting function that will use currently active counting
mode. To open a file os.read_file(path string) [https://github.com/vlang/v/blob/fdfa564865331a2bda077164d804d6dc90b83498/vlib/os/os.v#L96] is used which returns the file contents
and also closes the file. Nevertheless, we still need to ensure such a file
is present on the system with os.file_exists(_path string) [https://github.com/vlang/v/blob/fdfa564865331a2bda077164d804d6dc90b83498/vlib/os/os.v#L460].

fn count(mode Mode, path string) int {
    mut result := 0

    if !os.file_exists(path) {
        result = -1
        return result
    }

    content := os.read_file(path) or {return result}
    for item in content {
        if "" in mode.sep || item.str() in mode.sep {
            result++
        }
    }
    return result
}





There is one catch with the os.read_file(path string) [https://github.com/vlang/v/blob/fdfa564865331a2bda077164d804d6dc90b83498/vlib/os/os.v#L96] function, it returns an Option
type. This kind of type has to be handled in your code with an
or block that allows only specific set of keywords.

Once we handle the failing function and remove unnecessary printing to the
console the program is ready and complete.

Here is a challenge for you as a reader: Currently it handles only a single
file. Try to make it handle multiple files!

import os

struct Mode {
    name string
    cli_args []string
    sep []string
}


const (
    words = Mode{name: "words", cli_args: ["-w", "--words"], sep: [" ", "\n"]}
    lines = Mode{cli_args: ["-l", "--lines"], name: "lines", sep: ["\n"]}
    chars = Mode{name: "chars", cli_args: ["-c", "--chars"], sep: [""]}
)


fn parse_mode(args []string) Mode {
    mut mode := Mode{}

    if args[1] in words.cli_args {
        mode = words
    } else if args[1] in lines.cli_args {
        mode = lines
    } else if args[1] in chars.cli_args {
        mode = chars
    }
    return mode
}


fn count(mode Mode, path string) int {
    mut result := 0

    if !os.file_exists(path) {
        result = -1
        return result
    }

    content := os.read_file(path) or {return result}
    for item in content {
        if "" in mode.sep || item.str() in mode.sep {
            result++
        }
    }
    return result
}


fn main() {
    mode := parse_mode(os.args)
    file := os.args[2]
    println(count(mode, file))
}

Chapter IV: Role-Playing Game

In this chapter we’ll create a simplified role-playing game and leverage
multiple structures via struct [https://vlang.io/docs#structs]. Let’s start with place properties such as
name and links for connecting multiple place instances between themselves.
Each place will have one previous place and two places, one on the left side
and one on the right side.

               ___ <left>
              /
<previous> ---
              \___ <right>





To create a link between two place structures we need references to them and
use such references as values. A symbol & (ampersand) creates a reference
for a computer memory where our place struct [https://vlang.io/docs#structs] is stored. Reference is
a voidptr type.

struct Place {
    name string
    left &Place
    right &Place
    previous &Place
}





Each reference value for a struct [https://vlang.io/docs#structs] will start as a nil which is a pointer
that stores a value 0 and for that there is a checker in V - isnil(v voidptr) [https://github.com/vlang/v/blob/06a7954298a4200d7c774e2b28a7f4d0375d562c/vlib/builtin/builtin.v#L25].

Let’s try to create and connect these places with code:

                 ___ Pile of old leaves (left)
                /
Tree (start) ---
                \___ Shrubbery (right) ---
                                          \___ Bear behind Shrubbery (right)





Each of the place nodes can either have nil as a previous/next node or an
initialized different place. Set the reference again by using & character
before the symbol you want to reference, in this case an initialized place
struct [https://vlang.io/docs#structs].

fn main() {
    mut tree := Place{name: "Tree"}
    mut pile := Place{name: "Pile of old leaves"}
    mut shrub := Place{name: "Shrubbery"}
    mut bear := Place{name: "Bear behind Shrubbery"}

    // connect tree node with its children
    tree.left = &pile
    tree.right = &shrub
    pile.previous = &tree
    shrub.previous = &tree

    // forward-connect shrub node only
    // because it already has 'previous' set
    shrub.right = &bear
    bear.previous = &shrub

    println(&tree)
    println(tree)

    println(&pile)
    println(pile)

    println(&shrub)
    println(shrub)

    println(&bear)
    println(bear)
}





Running this piece of code won’t yet work due to struct [https://vlang.io/docs#structs] fields being
immutable. To mark them as mutable use mut keyword with
a colon (:) suffix. mut: will mark all fields after it as mutable
unless it encounters some other keyword changing the
struct [https://vlang.io/docs#structs] field properties.

struct Place {
    name string
mut:
    left &Place
    right &Place
    previous &Place
}





Once that is fixed, we get an output like below (the memory addresses will be
different). Notice the connections between each of the nodes such as Tree
memory address (0x7ffe01a7eac0 in this case) being equal Shrubbery previous
node’s address.

0x7ffe01a7eac0
{
    name: Tree
    left: 0x7ffe01a7ea90
    right: 0x7ffe01a7ea60
    previous: (nil)
}
0x7ffe01a7ea90
{
    name: Pile of old leaves
    left: (nil)
    right: (nil)
    previous: 0x7ffe01a7eac0
}
0x7ffe01a7ea60
{
    name: Shrubbery
    left: (nil)
    right: 0x7ffe01a7ea30
    previous: 0x7ffe01a7eac0
}
0x7ffe01a7ea30
{
    name: Bear behind Shrubbery
    left: (nil)
    right: (nil)
    previous: 0x7ffe01a7ea60
}





The places are ready, let’s create a traveler who will navigate through them.
Our traveler will for now contain a single property - location - which is
a reference for an already existing place. This always requires a value and
must not be nil. We can achieve such behavior by initializing the struct [https://vlang.io/docs#structs]
with a value and assigning only valid values to that property.

struct Traveler {
mut:
    location &Place
}

fn main() {
    ...
    traveler := Traveler{location: &tree}
    ...
    println(traveler.location.name)
}





While assigning invalid values e.g. an incorrectly typed (or raw) value is not
an error right now and we can happily do traveler.location = 0, it will
make the program access a part of memory which it should not be allowed to do,
therefore will result in a Segmentation fault (core dumped).

By using reference for a place and its references for the neighborhood creating
a function that will move the traveler is a piece of a cake. We just need to
watch out for nil with a short check that both accepts and returns
references for places.

fn nonnil_or_stay(old_place &Place, new_place &Place) &Place {
    if isnil(new_place) {
       println("Nothing is there.")
       return old_place
    } else {
       return new_place
    }
}





With this block we can repeat the check for multiple directions we would like
to travel - back, left or right - which helps with directly assigning them to
traveler’s location property since it should return either its old value or the
new one for the desired direction if not nil.

The function moving our traveler will require a mutable instance of traveler
struct [https://vlang.io/docs#structs] and an immutable string for the direction. Mutability can again be
achieved by mut keyword.

fn move(trav mut Traveler, direction string)





First pull some values out of the traveler instance directly and even from its
references into variables for easier access and less repetitive code.

place := trav.location
back := place.previous
left := place.left
right := place.right





Then complete the function by choosing the right location for traveler property
via the place checking function and stored place references,

fn move(trav mut Traveler, direction string) {
    place := trav.location
    back := place.previous
    left := place.left
    right := place.right

    println("Old location $trav.location.name")
    println("Trying to move $direction")

    if direction == "back" {
        trav.location = nonnil_or_stay(place, back)
    } else if direction == "left" {
        trav.location = nonnil_or_stay(place, left)
    } else if direction == "right" {
        trav.location = nonnil_or_stay(place, right)
    } else {
        println("$direction is not valid, use back, left or right")
    }
    println("New location $trav.location.name")
}





and replace the place properties’ values printing with move functions.

fn main() {
    ...
    // forward-connect shrub node only because it already has 'previous' set
    shrub.right = &bear
    bear.previous = &shrub

    move(mut traveler, "back")
    move(mut traveler, "left")
    move(mut traveler, "back")
    move(mut traveler, "right")
    move(mut traveler, "back")
}





Here is a sample output of how it should look like.

Old location Tree
Trying to move back
Nothing is there.
New location Tree
Old location Tree
Trying to move left
New location Pile of old leaves
Old location Pile of old leaves
Trying to move back
New location Tree
Old location Tree
Trying to move right
New location Shrubbery
Old location Shrubbery
Trying to move back
New location Tree





Any function you create with fn keyword can also contain
a special syntax before its name - a receiver - which is in simple terms just
a specification for the compiler to allow calling that function via dot-lookup
(using . for fetching an attribute of some entity).

Using (variable mutability Struct) syntax let’s make the function for
moving accessible directly from the traveler just by moving a function argument
slightly to the left in the function declaration. Also, don’t forget to make
the traveler mutable.

fn (trav mut Traveler) move(direction string) {
    ...
}

fn main() {
    ...
    mut traveler := Traveler{location: &tree}
    ...
    traveler.move("back")
    traveler.move("left")
    traveler.move("back")
    traveler.move("right")
    traveler.move("back")
}

Calculator test cases

>>> print(2 + 3)
5
>>> print(-2 + -3)
-5
>>> print((-2) + (-3))
-5
>>> print(2 + -3)
-1
>>> print(2 + (-3))
-1

>>> print(2 - 3)
-1
>>> print(-2 - -3)
1
>>> print((-2) - (-3))
1
>>> print(-2 - 3)
-5
>>> print((-2) - 3)
-5

>>> print(2 * 3)
6
>>> print(-2 * -3)
6
>>> print(-2 * 3)
-6

>>> print(2 / 3)
0
>>> print(-2 / -3)
0
>>> print(2 / -3)
0
>>> print(2 / 0)
.vrepl.v:2:11: division or modulo by zero

>>> print(2 / -1)
-2
>>> print(0 / 2)
0
>>> print(0 / -1)
0


>>> print(2.0 + 3)
5.000000
>>> print(-2.0 + -3)
-5.000000
>>> print((-2.0) + (-3))
-5.000000
>>> print(2.0 + -3)
-1.000000
>>> print(2.0 + (-3))
-1.000000

>>> print(2.0 - 3)
-1.000000
>>> print(-2.0 - -3)
1.000000
>>> print((-2.0) - (-3))
1.000000
>>> print(-2.0 - 3)
-5.000000
>>> print((-2.0) - 3)
-5.000000

>>> print(2.0 * 3)
6.000000
>>> print(-2.0 * -3)
6.000000
>>> print(-2.0 * 3)
-6.000000

>>> print(2.0 / 3)
0.666667
>>> print(-2.0 / -3)
0.666667
>>> print(2.0 / -3)
-0.666667
>>> print(2.0 / 0)
.vrepl.v:2:13: division or modulo by zero

>>> print(2.0 / -1)
-2.000000
>>> print(0.0 / 2)
0.000000
>>> print(0.0 / -1)
-0.000000


>>> print(2 + 3.0)
<garbage>
>>> print(-2 + -3.0)
<garbage>
>>> print((-2) + (-3.0))
<garbage>
>>> print(2 + -3.0)
<garbage>
>>> print(2 + (-3.0))
<garbage>

>>> print(2 - 3.0)
<garbage>
>>> print(-2 - -3.0)
<garbage>
>>> print((-2) - (-3.0))
<garbage>
>>> print(-2 - 3.0)
<garbage>
>>> print((-2) - 3.0)
<garbage>

>>> print(2 * 3.0)
<garbage>
>>> print(-2 * -3.0)
<garbage>
>>> print(-2 * 3.0)
<garbage>

>>> print(2 / 3.0)
<garbage>
>>> print(-2 / -3.0)
<garbage>
>>> print(2 / -3.0)
<garbage>
>>> print(2 / 0.0)
<garbage>
>>> print(2 / -1.0)
<garbage>
>>> print(0 / 2.0)
<garbage>
>>> print(0 / -1.0)
<garbage>

>>> print(2.3 + 3)
5.300000
>>> print(-2.3 + -3)
-5.300000
>>> print((-2.3) + (-3))
-5.300000
>>> print(2.3 + -3)
-0.700000
>>> print(2.3 + (-3))
-0.700000

>>> print(2.3 - 3)
-0.700000
>>> print(-2.3 - -3)
0.700000
>>> print((-2.3) - (-3))
0.700000
>>> print(-2.3 - 3)
-5.300000
>>> print((-2.3) - 3)
-5.300000

>>> print(2.3 * 3)
6.900000
>>> print(-2.3 * -3)
6.900000
>>> print(-2.3 * 3)
-6.900000

>>> print(2.3 / 3)
0.766667
>>> print(-2.3 / -3)
0.766667
>>> print(2.3 / -3)
-0.766667
>>> print(2.3 / 0)
.vrepl.v:2:13: division or modulo by zero
>>> print(2.3 / -1)
-2.300000
>>> print(0.3 / 2)
0.150000
>>> print(0.3 / -1)
-0.300000


>>> print(2 + 3.3)
<garbage>
>>> print(-2 + -3.3)
<garbage>
>>> print((-2) + (-3.3))
<garbage>
>>> print(2 + -3.3)
<garbage>
>>> print(2 + (-3.3))
<garbage>

>>> print(2 - 3.3)
<garbage>
>>> print(-2 - -3.3)
<garbage>
>>> print((-2) - (-3.3))
<garbage>
>>> print(-2 - 3.3)
<garbage>
>>> print((-2) - 3.3)
<garbage>

>>> print(2 * 3.3)
<garbage>
>>> print(-2 * -3.3)
<garbage>
>>> print(-2 * 3.3)
<garbage>

>>> print(2 / 3.3)
<garbage>
>>> print(-2 / -3.3)
<garbage>
>>> print(2 / -3.3)
<garbage>
>>> print(2 / 0.3)
<garbage>
>>> print(2 / -1.3)
<garbage>
>>> print(0 / 2.3)
<garbage>
>>> print(0 / -1.3)
<garbage>

>>> print(10 - (12 / 8) + 7 / 12 + 11)
20

>>> print(10.0 - (12.0 / 8.0) + 7.0 / 12.0 + 11.0)
20.083333

>>> print(8 + 2 - (11 / 4) * 6 + 3)
1

>>> print(8.0 + 2.0 - (11.0 / 4.0) * 6.0 + 3.0)
-3.500000

>>> print(2 * 2 + 11 * 4 - (10 + 6))
32

>>> print(2.0 * 2.0 + 11.0 * 4.0 - (10.0 + 6.0))
32.000000

>>> print(3 + (2 / 5 * 5 - 6 / 3))
1

>>> print(3.0 + (2.0 / 5.0 * 5.0 - 6.0 / 3.0))
3.000000

>>> print(7 - 10 * 2 - 4 * (6 - 9))
-1

>>> print(7.0 - 10.0 * 2.0 - 4.0 * (6.0 - 9.0))
-1.000000

>>> print(5 - (8 * 6 / 6 * 7) / 7)
-3

>>> print(5.0 - (8.0 * 6.0 / 6.0 * 7.0) / 7.0)
-3.000000

>>> print(6 + 9 / 10 - (10 / 6) + 3)
8

>>> print(6.0 + 9.0 / 10.0 - (10.0 / 6.0) + 3.0)
8.233333

>>> print(10 / (10 + 2 / 10) - 10 + 11)
2

>>> print(10.0 / (10.0 + 2.0 / 10.0) - 10.0 + 11.0)
1.980392

>>> print(7 * 3 / 8 - 5 + (5 * 9))
42

>>> print(7.0 * 3.0 / 8.0 - 5.0 + (5.0 * 9.0))
42.625000

>>> print(2 - (2 / 8 - 7 / 2 * 4))
14

>>> print(2.0 - (2.0 / 8.0 - 7.0 / 2.0 * 4.0))
15.750000

>>> print(7 + (6 * 5^2 + 3))
34

>>> import math
>>> print(7 + (6 * math.pow(5, 2) + 3))
<garbage>

>>> print(7.0 + (6.0 * math.pow(5.0, 2.0) + 3.0))
160.000000

>>> print(20 * 2 - (1/2) * 9.8 * 2^2)
<C error>  // should produce a negative int

>>> print(20.0 * 2.0 - (1.0/2.0) * 9.8 * 2^2)
>>> print(20.0 * 2.0 - (1.0/2.0) * 9.8 * 2.0^2.0)
>>> print(20.0 * 2.0 - (1.0/2.0) * 9.8 * (2.0^2.0))
.vrepl.tmp.c:5342: error: invalid operands for binary operation
// <error due to float ^ int or float ^ float>

>>> print(20 * 2 - (1/2) * 9.8 * (2^2)))
<garbage>  // should be 40

>>> print(20.0 * 2.0 - (1.0/2.0) * 9.8 * (2^2))
40.000000

>>> print(20.0 * 2.0 - (1.0/2.0) * 9.8 * math.pow(2, 2))
20.400000

>>> print(4^3^2)
5

>>> print(math.pow(4, math.pow(3, 2)))
262144.000000

>>> print(4^(3^2))
5

>>> print(math.pow(math.pow(4, 3), 2))
4096.000000






Warning

Watch out for incomplete symbols or uncasted types, it’ll result in garbage.

// int / int division results in an int
>>> print(1 / 3)
0

// int / float division results in an integer-like garbage
// different between each division
>>> print(1 / 3.)
14308928
>>> print(1 / 3.)
20928064
>>> print(1 / 3.0)
22931008
>>> print(1 / 3.0)
28653120

>>> print(1.0 / 3.0)
0.333333

  // helloworld.v
println("Hello, world!")  // with a new like character \n
print("Hello, world!")    // without a new like character

Keywords

As per the official V documentation [https://vlang.io/docs#keywords].


	fn - begins a function declaration in any of these shapes:


	Pure function without specified return type i.e. having the
default void return type.

fn name()
fn name(arg int, arg2 string, ...)







	Pure function with specified return type.

fn name() int
fn name(arg int, arg2 string, ...) int







	Struct method without and with argument and result types.

fn (m MyStruct) name() int
fn (m MyStruct) name(arg int, arg2 string, ...) int










For more examples and explanation check functions [https://vlang.io/docs#fns] and methods [https://vlang.io/docs#methods]
in V documentation.






	import - imports a module so a program can access it and its public
symbols such as constants, structs or functions


	mut - makes a variable mutable (editable)


	in - checks whether array [https://vlang.io/docs#arrays] or map [https://vlang.io/docs#maps] on the right side
contains a value on the left side (value in array, see in keyword [https://vlang.io/docs#in] for more)


	break - breaks an iteration of a for loop [https://vlang.io/docs#for]


	const - declares a module-level variable that will never change (see
const keyword [https://vlang.io/docs#consts] for more)


	return - returns a value from a function


	exit - exits a program when encountered, allows user to specify the exit
code for operating system


	continue - immediately skips to the next iteration of a loop


	break - stops the loop from any further iteration


	panic - exits a program and marks the program as unsuccessfully stopped
with an exit code.

Types

As per the Basic types (official V documentation) [https://vlang.io/docs#btypes].


Stores value


Numeric


	int - 32-bit integer


Warning

Current implementation falls back to pure C int type which is
not always 32-bit [https://stackoverflow.com/a/1231456/5994041].
See #2480 [https://github.com/vlang/v/issues/2480] for more.





	f32 - 32-bit floating-point number (see f32 implementation [https://github.com/vlang/v/blob/fdfa564865331a2bda077164d804d6dc90b83498/vlib/compiler/cheaders.v#L117])


	byte - 8-bit unsigned number, the smallest piece of memory which can
contain basic character set such as ASCII Character set [https://en.wikipedia.org/wiki/ASCII#Character_set] (see byte implementation [https://github.com/vlang/v/blob/14c273f273c2bd8bff1d0cb2f775ec755e43455c/vlib/compiler/cheaders.v#L115])


	bool - numeric type using false as negative value and true as
positive (currently inherited from C implementation)


	voidptr - type holding a numeric value of a location in computer memory
regardless its type or other properties, where user is responsible for the
correct casting (converting) to certain types or manipulation with the value
stored within the memory starting on that address - offsetting the stored
memory address up to specified data container size (or its multiples).

Considering x86 CPU and int type having 32-bit size, the starting value
could be something like 0x0000 and the data container being an int
would offset at 0x0004 or in other words 4 bytes.


Note

Void pointer does not have the knowledge of the offset required for
program to retrieve a value from ones and zeroes starting at the numeric
value stored in the void pointer itself i.e. void pointer does not store
the value’s type.










Textual


	string - A read-only, immutable, array of bytes encoded using UTF-8
(see string [https://vlang.io/docs#strings] for more)









References value


	array - structure holding pointers for a value or another pointer (see
array [https://vlang.io/docs#arrays] for documentation and array implementation [https://github.com/vlang/v/blob/f7c00b818074689b5d87c1c571f0de5832820a7e/vlib/builtin/array.v#L9] for implementation details)







Option type

As per the Option types & error handling (official V documentation) [https://vlang.io/docs#option].

This kind of type is created by a ? (question mark) prefix before an
ordinary type such as string [https://vlang.io/docs#strings] or an int resulting in a ?string
return type for a function. This allows you to either return a correct value
for this type or mark the result of a function as a failure by using
return none which forces the function caller to handle such a case with
either a compile error unhandled option type: `?type` when you try to
access the value or a compile error saying this type does not have a method
for performing a certain action.

For example:

import os
content := os.read_file(<path>)
println(content)





does not compile due to `?string` needs to have method `str() string` to be
printable, however a conversion to string [https://vlang.io/docs#strings] i.e. accessing the value
of ?string type falls back to the original unhandled option type error.

import os
content := os.read_file(<path>)
println(content.str())






Allowed keywords

By default an optional type is handled by an or block such as this:

import os
content := os.read_file(<path>) or {
    // handle function failure
}
println(content.str())





and it allows only specific set of keywords:


	return


	exit


	continue


	break


	panic

Word list

afford
ignorance
beam
abnormal
flu
emphasis
brown
compliance
jury
judicial
bulb
depend
marsh
syndrome
jaw
understanding
fortune
wage
cruel
wrong
outlet
presidential
average
make
trip
champion
high
advice
bargain
cup
constitution
onion
arrangement
temporary
ecstasy
disaster
extract
cutting
employee
horizon
day
railcar
immune
carpet
bloodshed
chimpanzee
explosion
admit
variable
attract

